# $Id: PKGBUILD 88008 2013-04-09 21:48:29Z eric $
# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Contributor: Tobias Powalowski <tpowa@archlinux.org>
# Contributor: red_over_blue,Ben <ben@benmazer.net>,Kritoke <kritoke@gamebox.net>

pkgname=qscintilla-qt3
pkgver=1.7.1
pkgrel=4
pkgdesc="A port to Qt3 of Neil Hodgson's Scintilla C++ editor class"
arch=('i686' 'x86_64')
url="http://www.riverbankcomputing.co.uk/software/qscintilla/intro/"
license=('GPL2')
depends=('qt3')
source=("http://downloads.sourceforge.net/pyqt/QScintilla-1.71-gpl-$pkgver.tar.gz")
md5sums=('dfa047b45d4f09ae8d4a8a431ba88a5a')

build() {
  cd "${srcdir}"/QScintilla-1.71-gpl-$pkgver
  cd qt
  sed -i "s%DESTDIR.*%DESTDIR=$pkgdir/usr/lib%g" qscintilla.pro
  qmake-qt3 qscintilla.pro
  sed -i 's|$(QTDIR)/include|/usr/include/qt3|' Makefile
  make MOC=/usr/bin/moc-qt3
}

package() {
  cd "${srcdir}"/QScintilla-1.71-gpl-$pkgver/qt
  # installind includes
  for i in qextscintilla*.h; do
    install -m 644 -D $i "${pkgdir}"/usr/include/qt3/$i
  done
}
